UPMC/licence/info/2I008
Fonctions et procédures

P. Manoury

Janv. 2017

1 Types de base et expressions typées

1.1 Type `bool`

Type des *valeurs booléennes*.
- 2 *constantes* : `true` et `false`
- 3 *opérateurs* : `&&`, `||` et `not`
  Application :
  - notation *préfixe* : `(not` e`)`
  - notations *infixe* : `(`e_1 `&&` e_2`)` et `(`e_1 `||` e_2`)`

Une *expression booléenne* est une expression *de type* `bool`.

Type des constantes et opérateurs :

```
true   :  bool
false  :  bool
not    :  bool -> bool
&&     :  bool -> bool -> bool
||     :  bool -> bool -> bool
```

Si e_1 et e_2 sont des expressions booléennes alors `true`, `false`, `(not` e_1`)`, `(`e_1 `&&` e_2`)` et `(`e_1 `||` e_2`)` sont des expressions booléennes, et réciproquement.

Expressions fonctionnelles La fonction `xor` («ou exclusif») n'est pas définie.
L'expression booléenne `((`e_1 `||` e_2`) && (not (`e_1 `&&` e_2`)))` a pour valeur `true` si et seulement si

1. e_1 a pour valeur `true` et e_2 a pour valeur `false`

   ou

2. e_1 a pour valeur `false` et e_2 a pour valeur `true`

L'*expression fonctionnelle*

`(`**`fun`** e_1 e_2 **`->`** `((`e_1 `||` e_2`) && (not (`e_1 `&&` e_2`))))`

a pour *valeur* la fonction `xor`. Elle est de type `bool -> bool -> bool`.

Une expression fonctionnelle est construite avec le *mot clé* **`fun`** et le *symbole réservé* **`->`**.

Dans l'expression fonctionnelle ci-dessus, e_1 et e_2 sont des *paramètres formels*. Les paramètres formels sont des *identificateurs* (noms de variables), ils peuvent varier. L'expression

```
(fun x y -> (x || y) && (not (x && y)))
```

dénote à la même fonction que `(`**`fun`** e_1 e_2 **`->`** `(`e_1 `||` e_2`) && (not (`e_1 `&&` e_2`)))`.

Application et évaluation L'application des expressions fonctionnelles utilise la notation préfixe. Pour obtenir la valeur d'une application, les paramètres formels sont remplacés par les *paramètres d'appel* lors de l'application des fonctions. Par exemple :

— l'application `((fun x y -> (x || y) && (not (x && y))) true false)` a pour valeur `true` ;

— l'application `((fun x y -> (x || y) && (not (x && y))) true true)` a pour valeur `false`.

Les paramètres d'appels sont n'importe quelle expression, pourvu qu'elle soit du type attendu (ici, `bool`).

La valeur de l'application `((fun x y -> (x || y) && (not (x && y))) true false)` est la valeur de l'expression `(x || y) && (not (x && y))` où `x` à la valeur `true` et `y` la valeur `false` ; c'est-à-dire, la valeur de l'expression `(true || false) && (not (true && false))`.

Définition En donnant un nom à une expression fonctionnelle, on définit une fonction. On donne un nom à une expression avec la construction `let`. Par exemple :

```
let xor = fun x y -> (x || y) && (not (x && y))
```

On peut vérifier que

— l'application `(xor true false)` a pour valeur `true` ;

— l'application `(xor true true)` a pour valeur `false`.

Sucre syntaxique pour les définitions de fonctions On peut écrire la définition

```
let xor x y = (x || y) && (not (x && y))
```

On lit cette définition comme : *«la valeur de l'application `(xor x y)` est égale à la valeur de l'expression `((x || y) && (not (x && y)))`, pour toute valeur de `x` et `y`.»*

L'expression à droite du symbole = est appelé le *corps de la fonction* définie.

Explicitation des types dans une définition de fonction On peut écrire, et on écrira

```
let xor (x:bool) (y:bool) : bool =
  (x || y) && (not (x && y))
```

Expression alternative C'est une expression qui correspond à la construction logique *«si ...alors ...sinon ...»*. Elle s'obtient en utilisant les mots clé `if`, `then` et `else`. Elle a la forme

$$\texttt{if}\ e\ \texttt{then}\ e_1\ \texttt{else}\ e_2$$

où e est une expression booléenne. Les expressions e_1 et e_2 peuvent avoir n'importe quel type, mais il faut que e_1 et e_2 soient du même type.

La valeur d'une expression alternative est définie ainsi :

— si e a la valeur `true` alors (`if` e `then` e_1 `else` e_2) a la valeur de e_1 ;

— si e a la valeur `false` alors (`if` e `then` e_1 `else` e_2) a la valeur de e_2.

On peut dire également qu'une expression alternative vérifie les équations :

$$(\texttt{if true then}\ e_1\ \texttt{else}\ e_2) = e_1$$
$$(\texttt{if false then}\ e_1\ \texttt{else}\ e_2) = e_2$$

Avec une expression alternative, on obtient une autre définition de la fonction `xor` :

```
let xor (b1:bool) (b2:bool) : bool =
  if b1 then (not b2)
  else b2
```

Filtrage Les langages ML offrent une construction particulière pour distinguer entre plusieurs *cas* de valeur d'une expression. C'est la construction de *filtrage*. Elle s'obtient avec les mots clé `match` et `with` ainsi que les symboles -> et `|`. On peut la rapprocher de la construction d'alternative généralisée de C ou JAVA : le `switch`. Mais elle est bien plus riche, comme nous le verrons par la suite.

Avec les booléens, son utilisation est simple et très proche de l'alternative car il n'y a que deux cas de valeurs pour une expression booléenne : `true` ou `false`. On peut l'utiliser pour définir de manière exhaustive, à la manière d'une table de vérité, la fonction `xor` :

```
let xor (b1:bool) (b2:bool) : bool =
  match b1, b2 with
      true, true -> false
    | true, false -> true
    | false, true-> true
    | false, false -> false
```

Pour évaluer une expression de filtrage, le résultat de l'évaluation des expressions analysées sous comparées séquentiellement avec les motifs de filtrage. La valeur de l'expression de filtrage est celle de la valeur qui se trouve à droite du symbole -> du premier motif qui correspond aux valeurs analysées.

On peut inverser l'ordre des motifs de filtrage lorsque les valeurs désignées sont exclusives les unes des autres. Par exemple

```
let xor (b1:bool) (b2:bool) : bool =
  match b1, b2 with
      true, false -> true
    | false, true-> true
    | false, false -> false
    | true, true -> false
```

Il existe, pour la construction de filtrage une espèce de «sinon» sous la forme d'un *motif universel* noté `_` (caractère «souligné»). Il permet de faire l'économie de comparaisons inutiles. Par exemple :

```
let xor (b1:bool) (b2:bool) : bool =
  match b1, b2 with
      true, false -> true
    | false, true-> true
    | _ -> false
```

Enfin, les motifs peuvent mentionner des noms de variables. Dans ce cas, la variable prend la valeur analysées correspondante et elle peut figurer dans l'expression se trouvant à droite du symbole `->` associé au motif. Par exemple :

```
let xor (b1:bool) (b2:bool) : bool =
  match b1, b2 with
      true, true -> false
    | true, false -> true
    | false, r -> r
```

ATTENTION : l'usage du filtrage est assez délicat. Il nécessite de savoir quel ensemble de valeurs analysées correspond à un motif pour placer ceux-ci dans un ordre adéquat.

1.2 Type `int`

Type des valeurs entières, positives ou négatives (entiers relatifs). On appelle *expression entière* les expressions de type `int`.

- Les constantes du type `int` peuvent être écrites en notation *décimale*. Par exemple : `-42`, `0` ou `42`. On peut aussi utiliser la notation *hexadécimale* en mentionnant le préfixe `0x`. Par exemple, 42 s'écrit `0x2a`. Il existe également une notation *binaire* des entiers, avec le préfixe `0b`. Par exemple, 42 s'écrit `0b101010`.
- Les opérateurs arithmétiques de base sont `+ - * / mod abs succ pred`. Les opératieurs binaires utilisent la notation infixe : `(`e_1 `+` e_2`)` `(`e_1 `-` e_2`)` `(`e_1 `*` e_2`)` `(`e_1 `/` e_2`)` `(`e_1 `mod` e_2`)`. Les opérateurs unaires, la notation préfixe : `(succ` e`)` `(pred` e`)` `(abs` e`)`.
  Attention au signe `-` qui est aussi utilisé comme opérateur de la soustraction. Il faut souvent utiliser des parenthèse pour désambigüer son utilisation :
  `(abs -1)` n'est pas correct ;
  `(abs (-1))` est correct.
- Les opérateurs «bit à bit» sont des opérateurs de bas niveau qui reposent sur la *représentation binaire* des entiers. On y trouve la opérateurs logiques : `land lor lxor lnot`, en notation infixe pour les trois premiers et préfixe pour `lnot`. On y trouve également les opérateurs de décalage : `lsr lsl asr` ; notation infixe.

L'ensemble des valeurs de type `int` est *fini*. Une valeur de type `int` est en effet contenue dans un *mot mémoire*. Le nombre de valeurs entière varie donc selon l'architecture de la machine où sont exécutés les programmes (32 bits ou 64 bits, par exemple). On a deux constantes prédéfinies qui donnent la valeurs du plus petit et du plus grans entiers représentable, respectivement : `min_int` et `max_int`.

Les opérations arithmétiques sont effectuées *modulo* cette limite de taille : `max_int + 1` a pour valeur celle de `min_int`.

Fonctions partielles et exception La division est une *fonction partielle* : elle n'est pas définie si le diviseur est égal à 0. L'expression entière `1/0` n'a pas de valeur. Elle est syntaxiquement correcte et bien typée (`1` et `0` sont des expressions entières, la division `/` est de type `int -> int -> int`, donc `1/0` est une expression entière). Mais son évaluation *déclenche l'exception* : `Division_by_zero`. Lorsqu'une exception est déclenchée, le programme est interrompu. L'exception signale ici une *erreur d'exécution*.

Opérateurs de comparaison et polymorphisme On peut utiliser avec les entiers les opérateurs de comparaison : < <= = >= > et <>, en notation infixe.

Toutefois, les opérateurs de comparaison ne sont pas propres aux entiers. En effet, si on peut comparer deux entiers (par exemple, `(0 = 1)` a la valeur `false`), on peut également comparer deux booléens : par exemple, `(false < true)` a la valeur `true`.

Ainsi, les opérateurs de comparaison sont des opérateurs *polymorphes*. Ils peuvent prendre en argument des valeurs de n'importe quel type pour peu que ce soit le même : on ne peut pas comparer un entier avec un booléen.

Pour noter le type des opérateurs polymorphe, on utilise des *variables de type*. Syntaxiquement, les variables de type sont reprérées par l'usage du symbole `'` (caractère «apostrophe»). Par exemple, les opérateurs de comparaison ont le type `'a -> 'a -> bool`. La mention répétée de la variable de type `'a` force l'identité de type entre les deux arguments des opérateurs de comparaison.

Fonctions récursives La fonction d'élévation à la puissance n'est pas définie. Il n'y a pas d'expression qui donne la valeur de «x à la puissance n». Mais :

$$\begin{array}{lcl}
\text{«}x \text{ à la puissance } 0\text{»} & = & 1 \\
\text{«}x \text{ à la puissance } 1\text{»} & = & x \\
\text{«}x \text{ à la puissance } 2\text{»} & = & x \times x \\
\text{«}x \text{ à la puissance } 3\text{»} & = & x \times x \times x \\
\text{etc.} & &
\end{array}$$

Schématiquement :

$$\text{«}x \text{ à la puissance } n\text{»} \quad \text{est égal à} \quad \underbrace{x \times \ldots \times x}_{n\ fois}$$

Il est alors crucial de faire la remarque suivante. Lorsque n est plus grand que 0 :

$$\underbrace{x \times \ldots \times x}_{n\ fois} \quad \text{est égal à} \quad x \times \underbrace{x \times \ldots \times x}_{n-1\ fois}$$

Cette remarque nous permet de poser *l'équation récursive suivante* :

$$\text{«}x \text{ à la puissance } n\text{»} = x \times \text{«}x \text{ à la puissance } (n-1)\text{»}$$

Et on a par convention que «x à la puissance 0» $= 1$. De cela on déduit la *définition récursive* :

$$\begin{array}{lcll}
\text{«}x \text{ à la puissance } n\text{»} & = & 1 & \text{, si } (n = 0) \\
 & = & x \times \text{«}x \text{ à la puissance } (n-1)\text{»} & \text{, sinon}
\end{array}$$

Écrivons $pow(x, n)$ pour «x à la puissance n». Les *équations récursives* ci-dessus deviennent :

$$\begin{array}{lcll}
pow(x, n) & = & 1 & \text{si } n = 0 \\
 & = & x \times pox(x, n-1) & \text{sinon}
\end{array}$$

On peut transcrire cette définition en Ocaml. Pour poser une *définition récursive*, on utilise les mots clé `let` et `rec`, l'un après l'autre :

```
let rec pow (x:int) (n:int) : int =
   if (n = 0) then 1
   else x * (pow x n)
```

Cette définition donne les schémas d'évaluations attendus :

`(pow` x `0)`	$=$	`1`
`(pow` x `1)`	$=$	x `* (pow` x `0)`
	$=$	x `* 1`
	$=$	x
`(pow` x `2)`	$=$	x `* (pow` x `1)`
	$=$	x `*` x `* (pow` x `0)`
	$=$	x `*` x `* 1`
	$=$	x `*` x
etc.		

Fonction partielle et exception La fonction `pow` est de type `int -> int -> int`. Comme `(-1)` est de type `int`, l'application `(pow` e `(-1))` est correcte, pour toute expression entière e. Mais,

`(pow` e `(-1))` $=$ e `* (pow` e `(-2))` $=$ e `*` e `* (pow` x `(-3))` $= \ldots$

La fonction `pow` est partielle en son second argument : il y a des arguments d'appel e_2 de type `int` pour lesquelles, quelle que soit l'expression entière e_1, l'application `(pow` e_1 e_2`)` n'a pas de valeur. En langage courant, on dit que «la fonction boucle», sous entendu, «infiniment».

Pour éviter ce cas de figure, on peut adopter une attitude de *programmation défensive* qui consiste à vérifier que le second argument ne risquera pas de provoquer une évaluation indéterminée. Pour la fonction `pow`, on vérifie que la valeur du second argument n'est pas strictement négative.

On a deux manières de mettre en œuvre la programmation défensive :

a) on complète le domaine de la fonction, en convenant, par exemple, que pour les puissances négatives, la valeur de l'application est 1. Ce qui donne la définition :

```
let rec pow (x:int) (n:int) : int =
  if (n <= 0) then 1
  else x * (pow x (n-1))
```

On aura ici que `(pow x -1)` a pour valeur 1. Cette première manière a toutefois l'inconvénient que l'application d'un argument impropre peu passer inapercue.

b) on intercepte les applications hors domaine en *déclenchant* une exception. L'utilisation d'un arguement impropre ne passera pas alors inaperçue.
On peut pour cela utiliser la *fonction prédéfinie* `failwith` :

```
let rec pow (x:int) (n:int) : int =
  if (n < 0) then failwith "invalid exponent"
  else if (n = 0) then 1
  else x * (pow x (n-1))
```

La fonction prédéfinie `failwith` déclenche l'exception `Failure` accompagnée d'un message sous forme de chaîne de caractère (voir **??** pour les chaînes de caractères).
On peut également utiliser la *fonction primitive* `raise` avec une exception prédéfinie du langage. L'exception `Invalid_argument` est d'ailleurs prévue à cet effet :

```
let rec pow (x:int) (n:int) : int =
  if (n < 0) then raise (Invalid_argument "pow: negative exponent")
  else if (n = 0) then 1
  else x * (pow x (n-1))
```

Définition locale (de fonction) Notre mise en œuvre de la programmation défensive pour la fonction `pow` à l'inconvénient d'avoir à évaluer à chaque appel récursif de la fonction un test qui n'est utile qu'une seule fois : lors du «premier» appel de la fonction. Pour palier cet inconvénient, on divise le calcul de la fonction en deux temps :

— Tester si le second argument est négatif ou non.

— Calculer récursivement l'élévation à la puissance, si le second argument n'est pas négatif. Dans ce cas, on pourra utiliser une définition «incomplète».

Pour réaliser cela, on utilise une *définition locale* de fonction. On obtient une définition locale avec les mots clé `let` (éventuellement suivi de `rec`) et `in`. Voici comment cela se présente dans notre cas :

```
let pow (x:int) (n:int) : int =
  let rec loop (n:int) =
    if (n = 0) then 1
    else x * (loop (n-1))
  in
    if (n < 0) then raise (Invalid_argument "pow: negative exponent")
    else (loop n)
```

Notez que la définition de `pow` elle-même n'est pas une définition récursive : on n'a pas écrit `let rec pow ...` La fonction récursive est la fonction locale `loop`. Celle-ci est utilisée dans l'expression qui vient après le mot clé `in`. L'ensemble syntaxique `let rec loop ...in ...` forme l'expression qui définit la fonction `pow`.

La forme générale d'une déclaration locale est `let` x `=` e_1 `in` e_2 (ou `let rec` x `=` e_1 `in` e_2). C'est une expression dont la valeur est la valeur de e_2 dans laquelle x a la valeur de e_1.

Portée des variables Il y a plusieurs remarques à formuler concernant la *portée* des variables dans notre dernière définition de `pow`.

— la déclaration locale de la fonction `loop` a pour portée l'expression qui vient après le mot clé `in`. C'est-à-dire que dans cette expression, le nom `loop` est connu et peut être utilisé. Cette définition de `loop` ne peut être utilisée qu'à cet endroit : c'est sa *portée lexicale*. Aucune autre portion de programme ne peut utiliser cette définition.

— la fonction locale `loop` utilise la variable `x` alors que celle-ci ne fait pas partie des arguments de `loop`. Ceci est possible car la définition de `loop` est *dans la portée* des arguments de la définition de `pow`.

— enfin, l'argument de `loop` s'appelle `n`, comme le second argument de `pow`. Il ne peut toutefois pas y avoir ici de confusion pour le processus d'évaluation : la mention de `n` dans la définition de `loop` *masque* celle de `n` dans la définition de `pow`.

Récurrence terminale Le schéma d'évaluation de `(pow 5 3)` se développe ainsi :

```
(pow 5 3)  =  (loop 3)
           =  5 * (loop 2)
           =  5 * (5 * (loop 1))
           =  5 * (5 * (5 * (loop 0)))
           =  5 * (5 * (5 * 1))
           =  5 * (5 * 5)
           =  5 * 25
           =  125
```

On y observe deux phases :

la première s'achève lorsque les appels récursifs de `loop` ont tous été «résolus». dans cette phase, les multiplications par 5 sont «mises en attente».

la seconde consiste à effectuer les multiplications jusqu'à obtenir le résultat final.

On appelle ces deux phases, respectivement la *descente récursive*[1] et la *remontée récursive*. Dans certains cas, on peut fusionner ces deux phases en utilisant une forme de définition *récursive terminale*.

Pour transformer la définition de `loop` en forme récursive terminale, on ajoute à celle-ci un argument appelé *accumulateur*. Cet argument supplémentaire nous permettra d'effectuer les multiplications que notre première définition de `loop` laisse en attente. Voici comment les choses se présentent :

```
let pow (x:int) (n:int) : int =
  let rec loop (n:int) (r:int) =
    if (n = 0) then r
    else (loop (n-1) (x*r))
  in
    if (n < 0) then raise (Invalid_argument "pow: negative argument")
    else (loop n 1)
```

Si l'on développe le schéma d'évaluation de `(pow 5 3)`, on obtient à présent :

```
(pow 5 3)  =  (loop 3 1)
           =  (loop 2 5)
           =  (loop 1 25)
           =  (loop 0 125)
           =  125
```

1. Nous empruntons cette expression au domaine de l'analyse syntaxique.

L'accumulateur contient à chaque appel récursif une valeur correspondant au résultat des multiplications mises en attente dans la version non terminale. Lorsqu'il n'y a plus d'appel récursif, la valeur de l'accumulateur correspond bien à toutes les multiplications qu'il fallait effectuer. En général, on a que `(loop` n r`)` a pour valeur $5^n \times r$. Puisqu'à l'appel initial de `loop`, on donne à r la valeur 1, on obtient bien que `(loop` n `1)` a pour valeur $5^n \times 1$, soit 5^n.

Attention : nous avons pris la peine de stipuler que *«dans certains cas, on peut fusionner»* les phases d'évaluation de l'application d'une fonction récursive. Cela n'est pas toujours facile. Ce peut même devenir inutilement complexe.

Récurrence terminale et boucles Les définitions récursives servent à définir des fonctions par *itération* d'un calcul. Dans les langages impératifs, les *boucles* servent le même dessein. Nous illustrons brièvement ce rapport d'analogie.

Refoemulons la définition de la fonction `pow` en utilisant, pour la fonction récursibve `loop`, une *test de continuation* plutôt qu'un *test d'arrêt* :

```
let pow (x:int) (n:int) : int =
  let rec loop (n:int) (r:int) =
    if (n > 0) then (loop (n-1) (x*r))
    else r
  in
    if (n < 0) then raise (Invalid_argument "pow: negative exponent")
    else (loop n 1)
```

Dans cette version, la fonction `loop` est appelée récursivement «tant que» `(n > 0)` prend la valeur `true`.

Cela suggère la définition impérative suivante, avec le langage python :

```
def pow(x,n):
  def loop (n):
    r = 1
    while (n > 0):
      r = x * r
      n = n - 1
    return r
  if (n < 0):
    raise ValueError("pow: negative exponent")
  else:
    return loop(n)
```

Notez que dans cette version impérative, l'accumulateur n'est pas un argument de la fonction locale `loop`, mais une variable locale de celle-ci qui est modifiée, par affectation, dans la boucle `while`.

Fonctionnelle On peut définir une fonction qui réalise une «boucle récursive» générique contrôlée par une valeur entière. Étant donné une fonction f et une valeur a et un entier n, elle donnera la valeur de $\underbrace{(f \ldots (f}_{n\ fois}\ a) \ldots)$ – ce que l'on note aussi $f^n(a)$. On peut définir cette itération d'applications, récursivement, en fonction de n :

$$\begin{array}{rcll} f^n(a) & = & a & \text{si } n = 0 \\ & = & f(f^{n-1}(a)) & \text{sinon} \end{array}$$

Ce qui donne la définition Ocaml :

```
let rec iter (n:int) (f: 'a -> 'a) (a:'a) : 'a =
  if (n > 0) then (f (iter (n-1) f a))
  else a
```

On peut aussi poser les équations récursives suivantes :

$$\begin{array}{rcll} f^n(a) & = & a & \text{si } n = 0 \\ & = & f^{n-1}(f(a)) & \text{sinon} \end{array}$$

Ce qui donne la définition Ocaml :

```
let rec iter (n:int) (f: 'a -> 'a) (a:'a) : 'a =
  if (n > 0) then (iter (n-1) f (f a))
  else a
```

qui est récursive terminale.

La fonction `iter` a deux particularités :
- la fonction `iter` est de type : `int -> ('a -> 'a) -> 'a -> 'a`. C'est une fonction polymorphe ;
- c'est une *fonction d'ordre supérieure*, appelée aussi *fonctionnelle*, car l'un de ses arguments est lui même une fonction (`f: 'a -> 'a`).

Cette fonction permet de donner une définition compacte de l'élévation à la puissance. En effet, pour obtenir la valeur de «x à la puissance n», il faut itérer n fois la fonction «multiplier par x» sur la valeur neutre 1.

En utilisant une expression fonctionnelle comme argument de `iter`, on peut poser :

```
let pow (x:int) (n:int) : int =
  if (n < 0) then raise (Invalid_argument "pow")
  else (iter n (fun r -> x * r) 1)
```

où l'expression fonctionnelle `(fun r -> x * r)` représente l'opération «multiplier par x» ; elle est de type `int -> int`.

1.3 Type `float`

Type des «nombres à virgule». On appelera *flottant* les valeurs du type `float`.

Les constantes utilisent la *notation pointée*. Par exemple : `...-6.2832 ...-0.1 ...0.0 ...0.1 ...3.1416 ...`. On peut aussi utiliser une *exponentiation* (puissances de 10). On aura `56789e+3` a pour valeur 5678.9 ou `789e-3` a pour valeur 0.789.

Les opérateurs arithmétiques sont ceux que l'on a sur les entiers avec une petite différence de notation : les symboles sont suivi du caractère «point» : `+. -. *. /.`. Ils sont de type `float -> float -> float`.

Les types numériques `int` et `float` sont incompatibles :
l'expression `1 * 0.5` n'est pas correctement typée ;
l'expression `1.0 *. 0.5` est correctement typée.

Pour calculer avec des entiers ou des flottants on utilise des primitives de conversion explicites :
`int_of_float : float -> int` donne la partie entière de son argument ;
`float_of_int : int -> float`.

Il existe d'autres fonctions primitives ou prédéfinie sur les flottants, telles, par exemple, les fonctions trigonométriques. Nous n'en dirons pas plus sur ce sujet et vous invitons à consulter la documentation du langage.

1.4 Type `char`

Type des caractères ASCII 8 bits (ISO 8859-1).

Les constantes sont au nombre de 256. On les dénote en utilisant des apostrophes. Par exemple : `'0'` `...'9'` `...'A'` `...'Z'` `...'a'` `...'z'`. On peut également donner leur code ASCII : de `'\000'` à `'\255'`. Certains caractères ont des notations particulières, comme `'\n'` (retour-chariot) ou `'\t'` (tabulation).

Les opérateurs de codage et décodage des caractères sont fournis en standard :

— `int_of_char: char -> int` ;

— `char_of_int: int -> char`. Cette fonction est partielle et peut déclencher l'exception `Invalid_argument "char_of_int"`.

Exemple : décalage circulaire d'un caractère :

```
let shift_char (c:char) (d:int) : char =
  (char_of_int (((int_of_char c) + d) mod 255))
```

Exercice : décalage circulaire sur la plage `'!'` ...`'~'`

2 Structures linéaires

2.1 Type `string`

Types des *chaînes de caractères*. Une chaîne de caractères est une suite consécutive en mémoire de caractères. C'est une *structure de donnée linéaire*. Elle contient en fait une ensemble de valeurs qui sont des caractères.

Les constantes sont dénotées en utilisant des guillements anglais (caractère `"`). Par exemple : la *chaîne vide* s'écrit `""` ; on a aussi `"hello"`, `"hello\n"` qui est la même valeur que `"hello\010"`, etc.

L'opérateur de base sur les chaînes de caractères est la *concaténation* : symbole `^`, en notation infixe, de type : `string -> string -> string`. Par exemple, l'expression `("hello"^" "^"world!")` a pour valeur `"hello world!"`.

Le module `String` fournit une *bibliothèque* de fonctions sur les chaînes. Il fait partie de la *bibliothèque standard* du langage. Les fonctions de ce module sont accessibles par *notation pointée* : `String.nom_de_la_fonction`. On y trouve, entre autres :

`String.length: string -> int`
longueur de la chaîne, c'est-à-dire, le nombre de caractères qu'elle contient.

`String.get : string -> int -> char`
`(String.get str i)` donne le caractère en position `i` dans `str`. Le premier caractère est à la position 0. Abréviation : `str[i]` est un *alias* `(String.get str i)`. Fonction partielle : `Invalid_argument "index out of bounds"`

`String.index : string -> char -> int`
position d'un caractère dans une chaîne (1ère occurrence). Fonction partielle : `Exception: Not_found`.

`String.make : int -> char -> string`
construit une chaîne de longueur donnée avec le caractère donné

`String.sub : string -> int -> int -> string`
`(String.sub str pos len)` donne (une copie de) la sous-chaîne de `str` de longueur `len` qui commence à la position `pos`. Fonction partielle : `Exception: Invalid_argument "String.sub / Bytes.sub"` [2]

2. Le module `Bytes` est le module des chaînes *modifiables*, nous y reviendrons en ??

`String.map : (char -> char) -> string -> string`
chaîne obtenue par application d'une fonction sur chaque caractère. Nous donnons un exemple d'utilisation ci-dessous.

`String.mapi : (int -> char -> char) -> string -> string`
analogue à `String.map` avec une fonction qui prendra aussi en argument la position du caractère.

etc.

Schéma d'application Les fonctions `String.map` et `String.mapi` sont des exemples de fonctionnelles souvent utilisées avec les structures linéaires. Elles «transforme» une structure en appliquant une même fonction à chacun des éléments de la structure passée en argument.

Par exemple, un système de cryptage simple des messages est le *code de César.* qui consiste à décaler de manière circulaire, les lettres de l'alphabet.

En utilisant la fonction `shift_char` définie précédemment (**??**) et l'itérateur `String...map`, on définit :

```
let code_cesar (str:string) (d:int) : string =
  String.map (fun c -> (shift_char c d)) str
```

Pour décoder, prendre l'inverse du décalage (décalage négatif).

On obtient un cryptage un peu moins naïf en faisant dépendre le décalage de la position du caractère dans le message. La fonction `String.mapi` permet de définir :

```
let encode_cesari (str:string) : string =
  String.mapi (fun i c -> (shift_char c i)) str

let decode_cesari (str:string) : string =
 String.mapi (fun i c -> (shift_char c (-i))) str
```

Parcours récursif d'une chaîne par ses indices.

Pour vérifier l'intégrité d'une donnée, on calcule une *somme de contrôle* en additionnant les valeurs des octets qui constituent la donnée. Pour une chaîne de caractères, on additionne les codes ASCII des caractères. On réduit chaque étape du calcul *modulo 256* pour obtenir un octet de contrôle. Voici une méthode suggérée dans `https://fr.wikipedia.org/wiki/Somme_de_contrôle`.

```
let checksum1 (str:string) : int =
  let len = String.length str in
  let rec loop i =
    if (i < len) then (int_of_char str.[i])+(loop (i+1)) mod 256
    else 0
  in
    (loop 0)

let checksum2 (str:string) : int =
  let len = String.length str in
  let rec loop i =
    if (i < len) then (((int_of_char str.[i])*i)+(loop (i+1))) mod 256
    else 0
  in
    (loop 0)
```

Le parcours de la chaîne est réalisé par la fonction récursive locale `loop`. On a utilisé la déclaration locale `let ...in ...` pour mémoriser la longueur de la chaîne.

La somme de contrôle est formée de la chaîne contenant les deux octets de contrôle calculés avec `checksum1` et `checksum2` :

```
let string_of_char (c:char) : string =
  (String.make 1 c)

let checksum (str:string) : string =
  let c1 = char_of_int (checksum1 str) in
  let c2 = char_of_int (checksum2 str) in
    (string_of_char c1)^(string_of_char c2)
```

Schéma d'accumulation Le calcul des octets de contrôle répond à un *schéma d'accumulation* sur une structure linéaire. On peut implanter ce schéma sous la forme de la fonctionnelle suivante :

```
let string_foldi (str: string) (a:'a) (f: int -> char -'a -> 'a) : 'a =
  let len = String.length str in
  let rec loop i r =
    if (i < len) then (loop (i+1) (f i str.[i] r))
    else a
  in
    (loop 0 a)
```

On utilise `string_fold` pour définir, par exemple, `checksum2`

```
let checksum2 str =
  (string_foldi str 0 (fun i c r -> (int_of_char c)*i + r))
```

2.2 Type `('a list)`

Type des listes *paramatrées*. Les listes peuvent recevoir des valeurs appartenant à n'importe quel type de données. C'est en ce sens que le type des listes est *«paramétré»*. On peut ne pas connaître le type des éléments d'un liste et on utilise alors la notation d'un *type polymorphe* : `('a list)` où apparaît une variable de type, ici ; `'a`.

Les listes sont des *structures linéaires dynamiques*. Les chaînes de caractères sont des structures *statiques* : une fois crées, leur taille ne change plus. Pour ajouter un caractères à une chaîne, il faut en créer une nouvelle pour y recopier les éléments de l'ancienne chaîne plus celui, que l'on veut ajouter. L'ajout d'un élément dans une listes sera beaucoup plus économique ; si l'on ajoute en début de liste.

Constantes Il y a potentiellement une infinité de listes. E ce, à deux titres :
— la taille d'une liste n'est *a priori* pas limitée ;
— il y a autant de liste d'une taille donnée qu'il y a de types de valeurs.
La notation pour les constantes de listes utilise les crochets `[` et `]` ainsi que le point virgule `;`. La *liste vide* qui ne contient aucun élément est notée `[]`. On peut former des listes de booléens : `[true]`, `[true,false]`, `[true,false,false],...` ; des listes d'entiers : `[1]`, `[1;2]`, `...` ; de chaînes de caractères : `["hello"]`, `["hello"; "world"]`, `...` ; ou encore des listes de listes d'entiers : `[[]]`, `[[1]]`, `[[1],[1,2]]`, `...`, etc. Les potentialités sont infinies. Les seules contraintes sont physiques : la mémoire de la machine doit être suffisante pour contenir la structure ; logique : tous les éléments d'une liste appartiennent au même type. Les types de ces valeurs sont compètement déterminé. On

a, dans nos exemples, respectivement : `(bool list)`, `(int list)`, `(string list)` et `((int list) list)`.

Opérateur L'opérateur privilégié des listes est, à l'instar des chaînes de caractères, la concaténation. Elle est notée `@`, en infixe. Elle est de type `'a list -> 'a list -> 'a list`. Elle satisfait les schémas d'équation suivants :

```
[] @ [x1; ..; xn]              =  [x1; .., xn]
[x1; ..; xn] @ []              =  [x1; ..; xn]
[x1; ..; xn] @ [y1; ..; ym]    =  [x1; ..; xn; y1; ..; ym]
```

Constructeurs Les *constructeurs* des listes sont des opérations distinguées sur les listes. Ce sont les *primitives* qui sont à la base de la construction de toutes les structures de listes. Nous verrons, par exemple, comment on les utilisent pour définir la concaténation.

Le type des listes possède deux constructeurs. Le premier est une constante ; la liste vide que l'on note `[]`. Le second est l'opérateur qui permet d'ajouter un élément en tête de liste. Il est symbolisé par `::`, en notation infixe. Si `x` est une valeur, d'un certain type, et `xs` une liste d'éléments de cette valeur alors `x::xs` représente la liste qui commence par `x` et se poursuit par les éléments de `xs`. On prononce `x::xs` : `x` *«conse»* `xs`. Exemples :

— l'expression `x1::[]` a pour valeur celle de `[x1]` ;
— l'expression `x2::x1::[]` a pour valeur celle de `x2::[x1]` qui a pour valeur celle de `[x2; x1]`
— l'expression `x3::x2::x1::[]` a pour valeur celle de `x3::x2::[x1]` qui a pour valeur celle de `x3::[x2; x1]` qui, elle même a pour valeur celle de `[x3; x2; x1]`.

Ainsi toute expression de type `('a list)` s'évalue soit vers la constante `[]`, soit vers une structure *de la forme* `x::xs`.

Listes et filtrage La *structure de contrôle* `match-with` qui nous avait permis de distinguer entre plusieurs cas de valeurs des booléens peut également servir à distinguer les *cas de construction* des listes. Par exemple, on peut l'utiliser pour définir une fonction booléenne valant `true` si et seulement si son argument est une liste vide :

```
let is_empty (xs : 'a list) : bool =
  match xs with
    [] -> true
  | x::xs' -> false
```

On peut ici ne pas exprimer le deuxième cas, en utilisant le *motif universel* :

```
let is_empty (xs : 'a list) : bool =
  match xs with
    [] -> true
  | _ -> false
```

Définition récursive de la fonction de concaténation sur les listes.

La fonction de concaténation des listes satisfait les deux schémas d'équations suivants :

```
[] @ [y1; ..; ym]                =  [y1; .., ym]
[x1; x2; ..; xn] @ [y1; ..; ym]  =  [x1; x2; ..; xn; y1; ..; ym]
```

Appelons `ys` la liste notée `[y1; ..; ym]`. La première équation dit que : pour toutes liste `ys`, l'expression `[]@ys` a pour valeur celle de `ys` :

`[] @ ys` $=$ `ys`

Pour la seconde équation : remarquons que la liste notée `[x1; x2; ..; xn]` est égale à `x1::[x2; ..; xn]` ; de même, la liste notée `[x1; x2; ..; xn; y1; ..; ym]` est égale à `x1::[x2; ..; xn; y1; ..; ym]`. La seconde équation peut donc se réécrire :

`(x1::[x2; ..; xn]) @ [y1; ..; ym]` $=$ `x1::[x2; ..; xn; y1; ..; ym]`

Appelons `xs` la liste notée `[x2; ..; xn]` (attention : elle comence avec `x2`). Ici, intervient la remarque essentielle : la liste notée `[x2; ..; xn; y1; ..; ym]` correspond à la concaténation des listes `[x2; ..; xm]` et `[y1; ..; ym]`, c'est-à-dire à la valeur de l'expression `[x2; ..; xm] @ [y1; ..; ym]`. Ainsi, en utilisant les noms `xs` et `ys`, la seconde équation s'écrit :

`(x1::xs) @ ys` $=$ `x1::(xs @ ys)`

En résumé, la concaténation est définie par les deux équations

$$\left\{\begin{array}{lcl} \texttt{[] @ ys} & = & \texttt{ys} \\ \texttt{(x::xs) @ ys} & = & \texttt{x::(xs @ ys)} \end{array}\right.$$

Ces deux équations nous donnent une définition par cas de constructeur de la fonction de concaténation des listes. C'est une définition récursive puisque l'opération définie apparaît à gauche et à droite dans la seconde équation. Cette définition équationnelle est implémentée dans la bibliothèque standard de OCAML (module `Pervasives`) de la manière suivante :

```
let rec ( @ ) l1 l2 =
  match l1 with
    [] -> l2
  | hd :: tl -> hd :: (tl @ l2)
```

Le module `List` de la bibliothèque standard contient un nombre importants de fonctions utilitaires sur les listes. Elles sont en général de type polymorphe sur les éléments des listes. Citons :

`List.length : 'a list -> int` qui donne le nombre d'éléments de son argument, sa *longueur*.

`List.mem : 'a -> 'a list -> bool` qui donne la valeur `true` si et seulement si son premier argument appartient à la lsite donnée en second argument.

`List.nth : 'a list -> int -> 'a` telle que `(nth xs i)` donne l'élémént en `i`-ème position dans la liste `xs`, s'il existe. Le premier élément est à la position 0. On obtient l'exception `Failure "nth"` si `i` est supérieur ou égal à la longueur de `xs` ou l'exception `Invalid_argument "List.nth"` si `i` est négatif.

`List.rev : 'a list -> 'a list` telle que `(List.rev [x1; ..; xn])` donne la liste `[xn; ..; x1]`.

etc.

À titre d'exemples, voici les définitions de ces fonctions.

La fonction `length` : calculer la longueur d'une liste, c'est compter son nombre d'éléments. Si l'on considère les deux cas de construction des listes, on a :
— la liste `[]` ne contient aucun élément ;
— la liste `x::xs` contient un élément de plus que la liste `xs`.
La fonction `length` satisfait donc les deux équations :

$$\left\{\begin{array}{lcl} \texttt{(length [])} & = & 0 \\ \texttt{(length (x::xs))} & = & 1 + \texttt{(length xs)} \end{array}\right.$$

D'où la définition :

```
let rec length (xs : 'a list) : int =
  match xs with
    [] -> 0
  | _::xs -> 1+(length xs)
```

Comme on n'a pas eu besoin ici de la valeur du premier élément, on a utilisé le motif universel (_).

Cette fonction admet la définition récursive terminale suivante :

```
let length (xs : 'a list) : int =
  let rec loop (xs : 'a list) (r:int) : int =
    match xs with
      [] -> r
    | _::xs -> (loop xs (r+1))
  in
    (loop xs 0)
```

La fonction `mem` : pour déterminer si un élément `z` appartient ou non à une liste, on raisonne par cas de construction de la liste :

- si la liste est vide, alors `z` n'appartient pas à la liste ;
- si la liste est de la forme `x::xs` alors, il y a deux possibilités :
  - le premier élément de la liste est égal à `z` et donc `z` appartient à `x::xs`
  - `z` appartient à la liste `xs`

On peut donc poser que `mem` satisfait les *équations conditionnelles* suivantes :

$$\left\{\begin{array}{lcll} \texttt{(mem z [])} & = & \texttt{false} & \\ \texttt{(mem z (x::xs))} & = & \texttt{true} & \text{si } \texttt{x} = \texttt{z} \\ \texttt{(mem z (x::xs))} & = & \texttt{(mem z xs)} & \text{sinon} \end{array}\right.$$

D'où la définition :

```
let rec mem (z:'a) (xs:'a list) : bool =
  match xs with
    [] -> false
  | x::xs -> if (x=z) then true else (mem z xs)
```

On peut également dire que `z` est un élément de `x::xs` si et seulement si `x`=`z` ou `z` est un élément de `xs`. Ce qui donne la définition :

```
let rec mem (z:'a) (xs:'a list) : bool =
  match xs with
    [] -> false
  | x::xs -> (x=z) || (mem z xs)
```

La fonction (partielle) `nth` : quoique de réalisation assez simple, l'élaboration de cette fonction réclame un peu d'attention.

Intuitivement, l'application `(nth [x0; ..; xn] i)` a pour valeur `xi`, si `i` est compris entre 0 et n. Pour `i` négatif ou strictement supérieur à n, l'application n'a pas de valeur ; de même, `(nth [] i)` pas de valeur.

On peut remarquer que :

- l'indice du premier élément d'une liste est 0, donc l'application `(nth [x0; ..; xn] 0)` a pour valeur `x0` ;

— si l'indice d'un élément dans la liste [x1; ..; xn] est i alors, il est $i+1$ dans la liste [x0; x1; ..; xn] ; c'est-à-dire, si un élément est d'indice i dans la liste [x0; x1; ..; xn], il est d'indice i dans la liste [x1; ..; xn].

De ces remarques, on déduit les équations conditionnelles suivantes

(nth (x::xs) i)	=	x	si i=0
(nth (x::xs) i)	=	(nth xs (i-1))	sinon

On obtient de cette première analyse la définition suivante qui déclenche une exception lorsque la valeur n'est pas définie :

```
let rec nth (xs:'a list) (i:int) : 'a =
  match xs with
    [] -> (failwith "nth")
  | x::xs -> if (i=0) then x else (nth xs (i-1))
```

Si l'on regarde la spécification de la fonction List.nth, il y a en fait deux cas où la fonction n'a pas de valeur : le cas où l'indice est supérieur ou égal à la longueur de la liste et le cas où l'indice est négatif. Le premier cas est signalé par l'exception (Failure "nth") ; le second, par l'exception (Invalid_argument "List.nth").

On peut intercepter le second cas avant d'activer la recherche récursive par l'application d'une fonction locale, comme nous l'avions fait, par exemple, pour la fonction pow (??). Cela donne :

```
let nth (xs:'a list) (i:int) : 'a =
  let rec loop (xs:'a list) (i:int) : 'a =
    match xs with
      [] -> (failwith "nth")
    | x::xs -> if (i=0) then x else (loop xs (i-1))
  in
    if (i < 0) then raise (Invalid_argument "List.nth")
    else (loop xs i)
```

Le déclenchement de l'exception (Failure "nth") est cohérent avec la spécification : « On obtient l'exception Failure "nth" si i est supérieur ou égal à la longueur de xs ». En effet si, i étant positif, la succession d'appels récursifs atteint la liste vide sans avoir annulé i, puisque l'indice i est décrémenté de 1 à chaque appel récursif, c'est que l'indice était supérieur à la longueur de la liste.

La fonction rev : schématiquement, l'application (rev [x0; x1; ..; xn]) a pour valeur [xn; ..; x1; x0]. On peut remarquer que :

— (rev []) a pour valeur [] ;
— la liste [xn; ..; x1] est la valeur de (rev [x1; ..; xn]) ;
— la liste [xn; ..; x1; x0] est la valeur de la concaténation de [xn; ..; x1] et [x0].

On peut donc poser les deux équations suivantes :

$$\begin{cases} \texttt{(rev [])} & = \texttt{[]} \\ \texttt{(rev (x::xs))} & = \texttt{(rev xs) @ (x::[])} \end{cases}$$

Cela donnerait la définition suivante :

```
let rec rev (xs:'a list) : 'a list =
  match xs with
    [] -> []
  | x::xs -> (rev xs) @ [x]
```

Cette définition n'est toutefois pas très satisfaisante. Illustrons le en déroulant les étapes d'évaluation de `(rev [x1; x2; x3; x4])` :

```
(1)   (rev [x1; x2; x3; x4])  =  (rev [x2; x3; x4]) @ (x1::[])
(2)                           =  ((rev [x3; x4]) @ (x2::[])) @ (x1::[])
(3)                           =  (((rev (x4::[])) @ (x3::[])) @ (x2::[])) @ (x1::[])
(4)                           =  ((((rev []) @ (x4::[])) @ (x3::[])) @ (x2::[])) @ (x1::[])
(5)                           =  ((([] @ (x4::[])) @ (x3::[])) @ (x2::[])) @ (x1::[])
(6)                           =  (((x4::[]) @ (x3::[])) @ (x2::[])) @ (x1::[])
(7)                           =  ((x4::([] @ (x3::[]))) @ (x2::[])) @ (x1::[])
(8)                           =  ((x4::(x3::[])) @ (x2::[])) @ (x1::[])
(9)                           =  (x4::((x3::[]) @ (x2::[]))) @ (x1::[])
(10)                          =  (x4::x3::([] @ (x2::[]))) @ (x1::[])
(11)                          =  (x4::x3::(x2::[])) @ (x1::[])
(12)                          =  x4::((x3::(x2::[])) @ (x1::[]))
(13)                          =  x4::x3::((x2::[]) @ (x1::[]))
(14)                          =  x4::x3::x2::([] @ (x1::[]))
(15)                          =  x4::x3::x2::(x1::[])
```

Les étapes (1) à (5) correspondent au développement de la fonction `rev` en terme de concaténations. Les étapes (6) à (15) sont les étapes d'évaluation des différentes concaténation mises en attentente. C'est ici que le bât blesse : chaque évaluation d'une concaténation reparcours des valeurs déjà envisagées ; par exemple, `x4` est repris 3 fois avant d'arriver à sa place définitive.

Il est donc intéressant ici d'envisager l'utilisation d'un accumulateur, et d'une fonction récursive locale, pour éviter la mise en attente de concaténations :

```
let rev (xs:'a list) : 'a list =
  let rec loop (xs:'a list) (r:'a list) =
    match xs with
      [] -> r
    | x::xs -> (loop xs (x::r))
  in
    (loop xs [])
```

Schémas d'itération Le module `List` contient également nombre de fonctionnelles qui correspondent à des schémas génériques de traitement ou d'exploration de listes.

Schéma d'application

`List.map : ('a -> 'b) -> 'a list -> 'b list` telle que `(List.map f [x1; ..; xn])` a pour valeur la liste `[(f x1); ..; (f xn)]`.

Schéma de filtrage

`List.filter : ('a -> bool) -> 'a list -> 'a list` telle que `(List.filter p xs)` donne la liste des éléments `xi` de `xs` pour lesquels `(p xi)` vaut `true`.

Schémas d'accumulation

`List.fold_right : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b` telle que `(List.fold_right f [x1; ..; xn] a)` donne la valeur de l'expression `(f x1 .. (f xn a)..)`.
`List.fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a` telle que `(List.fold_left f a [x1; ..; xn])` donne la valeur de l'expression `(f (.. (f a x1) ..) xn)`.

À titre d'exemples, voici les définitions de ces itérateurs :

```
let rec map (f : 'a -> 'b) (xs : 'a list) : ('b list) =
  match xs with
```

```
    [] -> []
  | x::xs -> (f x)::(map f xs)
let rec filter (p : 'a -> bool) (xs : 'a list) : ('a list) =
  match xs with
    [] -> []
  | x::xs -> if (p x) then x::(filter p xs)
             else (filter p xs)
let rec fold_right (f : 'a -> 'b -> 'b) (xs : 'a list) (b : 'b) : 'b =
  match xs with
    [] -> b
  | x::xs -> (f x (fold_right f xs b))
let rec fold_left (f : 'a -> 'b -> 'a) (a : 'a) (xs : 'b list) : 'a =
  match xs with
    [] -> a
  | x::xs -> (fold_left f (f a x) xs)
```

Notez que `fold_left` est récursive terminale.

Un itérateur pour les gouverner tous On peut (re)définir les itérateurs `List.map` et `List.filter` comme cas d'application de `List.fold_right`.

La fonction itérée par `List.map` est simplement le constructeur *cons* :

```
let map (f : 'a -> 'b) (xs : 'a list) : 'b list =
  List.fold_right (fun x r -> (f x)::r) xs []
```

La fonction itérée par `List.filter` sélectionne les éléments à retenir pour le résultat final :

```
let filter (p : 'a -> bool) (xs : 'a list) : 'a list =
  List.fold_right (fun x r ->  if (p x) then x::r else r) xs []
```

L'itérateur `List.fold_right` peut permettre une expression très compacte de certaines fonctions. Par exemple, on obtient la fonction de recherche de l'élément maximal d'une liste de la manière suivante :

```
let find_max (xs : 'a list) : 'a =
  match xs with
    [] -> raise (Invalid_argument "find_max")
  | x::xs -> List.fold_right max xs x
```

Notez que `find_max` est une fonction partielle, non définie pour la liste vide.

On peut comparer cette définition à une définition plus directe :

```
let find_max (xs : 'a list) : 'a =
  let rec loop (xs : 'a list) (r : 'a) =
    match xs with
      [] -> r
    | x::xs -> (loop xs (max x r))
  in
    match xs with
      [] -> raise (Invalid_argument "find_max")
    | x::xs -> (loop xs x)
```

On peut noter que la fonction locale `loop` est terminale récursive. Cela indique que, dans ce cas, on peut tout aussi bien utiliser l'itérateur `List.fold_left` :

```
let find_max (xs : 'a list) : 'a =
  match xs with
    [] -> raise (Invalid_argument "find_max")
  | x::xs -> List.fold_left max x xs
```

L'utilisation de `List.fold_right` ou `List.fold_left` est indiférent dès lors que la fonction itérée est associative et commutative. On peut alors préfèrer `List.fold_left` qui est récursive terminale.

`List.fold_left` accumule *«à l'envers»* : on peut définir la fonction `rev` directement avec `fold_left`, version récursive terminale :

```
let rev (xs:'a list) : ('a list) =
  List.fold_left (fun r x -> x::r) [] xs
```

Pour avoir un `map` récursif terminal, il faut *2 passes* :

```
let map (f:'a -> 'b) (xs:'a list) : ('b list) =
  List.fold_left (fun r x -> x::r) []
      (List.fold_left (fun r x -> (f x)::r) [] xs)
```

Itération non bornée Parmi les fonctionnelles fournies par la bibliothèque standard, il y a celle-ci :

`find : ('a -> bool) -> 'a list -> 'a` telle que `(find p xs)` donne le premier élément `x`de `xs` pour lequel `(p x)` prend la valeur `true`, s'il existe, et déclenche l'exception `Not_found`, sinon.

On pourrait définir cette fonction en utilisant l'itérateur `List.filter` : en effet, le premier élément de `xs` pour lequel `p` prend la valeur `true` et le premier élément du résultat de `(List.filter p xs)`, si ce résultat n'est pas la liste vide. On aurait :

```
let find (p : 'a -> bool) (xs : 'a list) : 'a =
  match (List.filter p xs) with
    [] -> raise Not_found
  | x::_ -> x
```

Notez comment ici, on analyse le résultat de l'application d'une fonction et non plus simplement la forme d'un argument.

Toutefois, cette manière de faire n'est pas très optimale. En effet, `List.filter` explore toute la liste pour construire un résultat dont on ne conserve que le premier élément. Il eût été plus optimal d'arrêter la recherche dès que le premier `x` de `xs` pour lequel `p` prend la valeur `true` a été trouvé. Ce qui donne la définition :

```
let rec find (p : 'a -> bool) (xs : 'a list) : 'a =
  match xs with
    [] -> raise Not_found
  | x::xs -> if (p x) then x else (find p xs)
```

Il existe deux *conditions d'arrêt* pour cette fonction :

1. La liste est vide, au quel cas la recherche a échoué.

2. La fonction `p` prend la valeur `true` pour le premier élément de la liste, auquel cas, ce premeir élément est le résultat recherché.

De surcroît, cette définition est maintenant récursive terminale.

Les itérateurs `List.map` et `List.filter` implémentent des *itération bornées* sur les listes. Le nombre d'étapes d'itérations (*i.e.* le nombre d'appels récursifs) de l'application de ces itérateurs est égale à la taille des listes traitées.

En revanche, les fonctions de recherche d'une valeur dans une structure sont des exemples d'itérations *non bornées*. C'est-à-dire, d'itérations qui n'ont pas nécessairement besoin d'explorer la totalité des structures traitées pour produire un résultat.

Contrôles implicites Dans la définition de `find`, la structure de contrôle `if-then-else` permet de stopper le calcul dès que l'élément cherché est trouvé. Les opérateurs booléens de disjonction et de conjonction ont un effet de contrôle similaire. En fait :

— le processus d'évaluation de l'expression `b1 || b2` est équivalent à celui de l'expression `if b1 then true else b2` ;

— le processus d'évaluation de l'expression `b1 && b2` est équivalent à celui de l'expression `if b1 then b2 else false`.

Ainsi, le second argument d'une disjonction (`||`) ou d'une conjonction (`&&`) n'est pas nécessairement évalué.

C'est pourquoi, malgré les apparences, les deux définitions suivantes des fonction `List.for_all` et `List.exists` de la bibliothèque standard réalisent des itérations non bornées.

```
let for_all (p : 'a -> bool) (xs : 'a list) : bool =
  List.fold_left (fun x r -> (p x) && r) true xs
let exists (p : 'a -> bool) (xs : 'a list) : bool =
  List.fold_left (fun x r -> (p x) || r) false xs
```

3 Structures arborescentes

L'ensemble des listes est défini à l'aide de deux *constructeurs* : la constante `[]` pour la liste vide et l'opérateur `::` pour une liste non vide. Les listes sont des *structures récursives* : le second argument du constructuer de liste `::` est lui-même une liste. Cette manière d'envisager les structures de données permet d'en définir de plus complexes.

3.1 Arbres binaires

En calquant la définition récursive des listes, on peut définir les structures d'arbres binaires étiquetés de la manière suivante :

— on se donne un arbre vide ;

— on construit un arbre en ajoutant un premier élément (la *racine*) à deux arbres.

Pour appliquer ce principe dans le langage, on se donne deux symboles :

— `Empty`, constante pour désigner l'arbre vide ;

— `Node`, opérateur ternaire pour désigner l'opération d'ajout d'une racine pour connecter deux arbres.

3.2 Définition de type

Se donner ces symboles revient à *définir un nouveau type* dans le langage. La clause de définition d'une type est `type`. On écrit :

```
type 'a btree =
  Empty
| Node of 'a * ('a btree) * ('a btree)
```

Cette définition déclare le nom du nouveau type (`btree`). Comme les listes, il s'agit d'un type paramétré (variable de type `'a`). Le constructeur `Node` réclame trois arguments : l'élément placé à la racine, de type `'a` et les deux arbres réunis dans la structure construite, donc de type `'a btree`. Le mot clé `of` indique, si nécessaire le type des argument du constructeur. S'il y en a plusieurs, leur types sont séparés par une étoile (caractère `*`).

Le type ainsi définit appartient à la catégorie des *types somme*[3]. Nous y reviendrons en ??.

Toute valeur appartenant au type `'a btree` est égale soit à `Empty` soit à la valeur d'une expression de la forme `Node(x,bt1,bt2)` où `bt1` et `bt2` sont des expressions de type `'a btree`.

Notez : la manière dont les arguments d'un constructeur sont réunis dans un *n-uplet* (ici, un triplet) placé entre parenthèses.

Par exemple l'expression

```
Node(1,
     Node(2, Empty, Empty),
     Node(3,
          Empty,
          Node(4, Empty, Empty)))
```

a pour la valeur l'arbre (de type `int btree`) que l'on peut dessiner ainsi :



Les carrés représentent les (sous)arbres vides.

3.3 Filtrage et récurrence

Le principe de programmation (récursive) avec les arbres ressemble à celui que l'on a utilisé pour les listes. Il repose sur l'analyse des formes possible d'un arbre à l'aide le la structure de contrôle `match-with`.

Illustrons le avec la définition de la fonction `btree_mem`, de type `'a -> 'a btree -> bool` telle que `(btree_mem x bt)` vaut `true` si `x` apparaît dans l'arbre `bt` et `false` sinon. C'est l'analogue de la fonction `List.mem` pour les listes. Les propriétés de `btree_mem` seront donc également similaires : un arbre binaire est comme une liste, sauf qu'un arbre binaire non vide a deux suites, là où une liste non vide n'en a qu'une.

3. La terminologie est assez fluctuante pour ces types. On les nomme égaement *type union (étiquetée)*, *type algébrique* et dans la communauté ML, on utilise le terme *type variants*.

Par analogie avec les listes, en suivant une analyse par acs de constructeur des arbres, on peut donc dire

- que `(btree_mem x Empty)` vaut `false`, pour tout `x`
- que `x` apparaît dans `Node(y,bt1,bt2)` si et seulement si `x` est à la racine de `Node(y,bt1,bt2)` (c'est-à-dire, `x`=`y`), ou `x` apparaît dans `bt1` ou `bt2`.

La fonction `btree_mem` doit donc satisfaire les équations suivantes :

$$\left\{\begin{array}{lcl} \texttt{(btree\_mem Empty)} & = & \texttt{false} \\ \texttt{(btree\_mem x (Node(y,bt1,bt2)))} & = & \texttt{(x=y) || (btree\_mem x bt1) || (btree\_mem x bt2)}. \end{array}\right.$$

D'où la définition :

```
let rec btree_mem (x:'a) (bt:'a btree) : bool =
  match bt with
    Empty -> false
  | Node(y, bt1, bt2) -> (x=y) || (btree_mem x bt1) || (btree_mem x bt2)
```

3.4 Récurrence terminale

Considérons la fonction `btree_size` qui donne la taille d'un arbre (son nombre d'éléments). C'est ici l'analogue de la fonction `List.length`. Par analogie avec les propriétés de `List.length`, on pose :

$$\left\{\begin{array}{lcl} \texttt{size Empty} & = & \texttt{0} \\ \texttt{(size (Node(x,bt1,bt2)))} & = & \texttt{1 + (size bt1) + (size bt2)} \end{array}\right.$$

Ce qui donne la définition :

```
let rec size (bt:'a btree) : int =
  match bt with
    Empty -> 0
  | Node(_, bt1, b2) -> 1 + (size bt1) + (size bt2)
```

Si l'on veut obtenir une définition récursive terminale de cette fonction, on peut tenter de s'inspirer de la définition récursive terminale de `List.length` en introduisant une fonction récursive locale `loop` munie d'une *accumulateur*. Mais la chose est ici plus compliquée. En effet, pour calculer la taille de `Node(x,bt1,bt2)`, il faut calculer la taille de `bt1` et la taille de `bt2`. Il faut donc nécessairement procéder à deux appels récursifs. Ce qui est contradictoire avec la notion de récurrence terminale.

On peut toutefois obtenir quelque chose en analysant comment une expression telle que `1 + (size bt1) + (size bt2)` est évaluée. La valeur de cette expression ne sera obtenue que lorsque `(size bt1)` et `(size bt2)` auront été «résolues». Pour ce, le mécanisme d'évaluation commencera par l'évaluation de `(size bt1)` [4] ce qui provoque la «mise en attente» de l'évaluation de l'addition (`1 + ..`) et de `(size bt2)`. On avait vu comment supprimer la mise en attente de l'addition en ajoutant un accumulateur. Pour supprimer la mise en attente de `(size bt2)`, il suffit de se souvenir de `bt2` afin de le traiter à son tour lorsque `bt1` aura été traité. Et pour cela, il suffit d'introduire un nouvel accumulateur qui contient une liste d'arbres à traiter. Le calul se tremine lorsque l'arbre à traiter est vide et la liste d'arbres en attente égalenet. Cette idée donnera :

```
let size (bt:'a btree) : int =
  let rec loop (bt:'a btree) (bts:('a btree) list) (r:int) =
    match bt with
      Empty -> (
        match bts with
          [] -> r
        | bt::bts -> (loop bt bts r)
```

4. Ce choix est arbitraire, mais faire l'autre choix ne change rien à l'argument.

```
      )
      | Node(_,bt1,bt2) -> (loop bt1 (bt2::bts) (r+1))
  in
    (loop bt [] 0)
```

Dans la fonction auxiliaire `loop` le premier argument (`bt`) est tout simplement le premier arbre à traiter. On peut faire l'économie de cet argument en l'intégrant dans la liste des arbres à traiter. la fonction est alors définie par récurrence sur la liste d'arbres à traiter, et par analyse, par filtrage, du premier élément de la liste. On obtient alors la définition suivante :

```
let size (bt:'a btree) : int =
  let rec loop (bts:('a btree) list) (r:int) =
    match bts with
      [] -> r
    | (Empty::bts) -> (loop bts r)
    | ((Node(x,bt1,bt2))::bts) -> (loop (bt1::bt2::bts) (r+1))
  in
    (loop [bt] 0)
```

Notez le filtrage «en profondeur» : sur la liste et sur son premier élément.

Ce que nous avons réalisé avec cette dernière version de `size` n'est rien d'autre que la gestion explicite dans notre fonction du mécanisme d'empilement induit par l'évaluation d'une fonction récursive, non terminale. Le gain, en terme de consomation mémoire, est assez faible, puisque nous construisons une liste là où le mécanisme d'évaluation utilise la pile des appels de fonctions.

3.5 Exception et contrôle

Le mécanisme des exceptions permet de les «déclencher» avec la primitive `raise`. Il offre également la possibilité de les intercepter avec la construction `try ... with ...`

L'évaluation de l'expression `1/0` provoque le déclenchement de l'exception `Division_by_zero`. Le programme dans lequel cette expression aura été évaluée sera stopé à moins qu'un traitement pour cette exception n'y ait été prévu. Pour traiter une exception, on utilise la constrcution `try-with`. Par exemple `try (1/0) with Division_by_zero -> Printf.eprint "Divison par zéro"`. Dans ce cas, le programme n'est pas interrompu, et un message est affiché sur la sortie en erreur (`stderr`).

On peut utiliser le mécanisme des exceptions à la manière dont on utilise une expression alternative pour contrôler le flot d'évaluation d'un programme. Par exemple, on cherche dans un arbre binaire un élément satisfait une certaine propriété (exprimée comme une fonction booléenne). Le schéma de ce processus est proche de celui de la fonction `btree_mem` à ceci près que la fonction de recherche peut échouer (fonction partielle). Voici comment on peut décrire ce schéma :

- si l'arbre est vide, la recherche échoue ;
- si l'arbre a la forme `Node(x,bt1,bt2)` :
  - si `x` satisfait la propriété, la recherche réussit ;
  - sinon, essayer de cherche dans `bt1` :
    - si cela réussit, on a terminé ;
    - sinon, chercher dans `bt2`.

Ce qui se transcrit comme la définition de la fonction suivante qui donne un élément de l'arbre en cas de réussite :

```
let rec search (p:'a -> bool) (bt:'a btree) : 'a =
  match bt with
    Empty -> raise Not_found
```

```
  | Node(x,bt1,bt2) ->
     if (p x) then x
     else (
       try (search bt1)
       with Not_found -> (search bt2)
     )
```

Avec cette fonction, on peut donner une autre définition de la fonction `btree_mem`. Par exemple :

```
let btree_mem (x:'a) (bt:'a btree) : bool =
  try (x = search (fun y -> x=y) bt)
  with Not_found -> false
```

3.6 Parcours en largeur

La stratégie de recherche implémentée dans la fonction `search` ci-dessus est dite *en profondeur*. La figure ci-dessous illustre la succession selon laquelle les éléments de l'arbre sont examinés :



Les noeuds de l'arbre sont numérotés dans l'ordre dans lequel ils sont examinés. Il est parfois utile d'examiner les éléments dans l'ordre suivant :



Une telle manière de parcourir la structure arborescente est dite *en largeur*.

On ne peut réaliser cela directement avec une définition récursive reposant sur la structure des arbres. Il faut, à l'instar de ce qui a été réalisé pour une version récursive terminale de `size`, gérer explicitement l'ensemble des sous-arbres restants à traiter. Mais, pour obtenir un parcours en largeur, l'ordre dans lequel les (sous)arbres à traiter sont mémorisé n'est pas celui des listes (qui implante en fait un *mécanisme de pile* : le dernier élément ajouté est le premier accessible). Il faut utiliser une structure de *file d'attente* : le premier élément ajouté est le premier accessible.

Les files d'attentes quoique cela soit très inefficace, on peut utiliser une structure de liste pour réaliser la gestion de files d'attente. On définit alors les fonctions de base des gestion des files d'attente :

- reconnaître qu'une file est vide :

```
let is_empty (xs:'a list) : bool =
  match xs with
```

```
    [] -> true
  | _ -> false
```

- ajouter un élément dans un file d'attente

```
let add (x:'a) (xs:'a list) : 'a list =
  xs@[x]
```

- obtenir le premier élément de la file d'attente (si il existe)

```
let top (xs:'a list) : 'a =
  match xs with
    x::_ -> x
  | _ -> raise Not_found
```

- supprimer le premier élément de la file d'attente

```
let pop (xs:'a list) : 'a list =
  match xs with
    _::xs -> xs
  | _ -> xs
```

Munis de ces utilitaires, on définit ainsi un parcours en largeur d'un arbre binaire

```
let bfsearch (p:'a -> bool) (bt:'a btree) : 'a =
  let rec loop (bts:('a btree) list) =
    if (is_empty bts) then
      raise Not_found
    else (
        match (top bts) with
          Empty -> (loop (pop bts))
        | Node(x,bt1,bt2) -> (
            if (p x) then x
            else (loop (add bt2 (add bt1 (pop bts))))
        )
    )
  in
    (loop (add bt []))
```

Le module `Queue` il existe une réalisation bien meilleure des files d'attentes dans le module `Queue` de la bibliothèque standard. Le type de données des files d'attentes défini par ce module est `'a Queue.t`.

Toutefois les fonctions fournies par ce module ne sont pas... fonctionnelles. Par exemple, la fonction d'ajout d'un élément dans une file d'attente a pour signature `Queue.add: 'a -> 'a Queue.t -> unit`. La valeur de retour de cette «fonction» n'est pas une file d'attente mais la valeur de type `unit` que l'on note `()`. Elle procède à l'ajout par *effet de bord*, par *modification en place* de la structure. On a donc affaire à une «fonction» qui fait usage des traits impératifs des langages de programmation comme l'affectation. On avait usage d'appeler de telles fonctions des *procédures*[5]. Dans la même veine, `Queue.top`, de type `'a Queue.t -> 'a` supprime l'élément en tête de la file **et** délivre également la valeur de cet élément.

Le caractère impératif de `Queue.add` nous interdit l'usage d'expressions telles que `(loop (Queue.add bt2 (Queue.add bt1 (pop bts))))`. Il faut remplacer la *composition* fonctionnelle d'expressions par leur mise en *séquence*. Là où l'on a écrit `(loop (add bt2 (add bt1 bts)))`, on écrit `(Queue.add bt1 bts); (Queue.add bt2 bts); (loop bts)`.

5. D'où le titre de ce cours :)

Avec l'utilisation des files d'attentes du module Queue, la fonction de recherche devient

```
let bfsearch (p:'a -> bool) (bt:'a btree) : 'a =
  let rec loop (bts: ('a btree) Queue.t) =
    if (Queue.is_empty bts) then
      raise Not_found
    else (
      match (Queue.pop bts) with
          Empty -> (loop bts)
        | Node(x,bt1,bt2) -> (
            if (p x) then x
            else (
              Queue.add bt1 bts;
              Queue.add bt2 bts;
              loop bts)
            )
    )
  in
  let bts = Queue.create () in
    Queue.add bt bts;
    (loop bts)
```

Par rapport à sa version purement fonctionnelle, on perd ici la *transparence référentielle* du code : dans la définition de loop, l'identificateur bts ne désigne pas toujours la même valeur.

3.7 Arbres généraux

Structure arborescente dont le nombre de branchements est variable



Autre représentation graphique



Dans cette dernière, chaque nœud est constitué d'un nombre fixé de composants :

1. L'étiquette (les carrés).
2. La suite des sous-arbres (les barres horizontales).

Pour réaliser ces suites, de taille variable, possiblement vides, on utilise la structure dynamique de listes. D'où la définition :

```
type 'a gtree =
  GEmpty
| GNode of 'a * ('a gtree) list
```

L'arbre donné en exemple est représenté par d'expression

```
GNode(1,
  [ GNode(2, [ GNode(5, []);
               GNode(6, [ GNode(11, [])]);
               GNode(7, []) ]);
    GNode(3, [ GNode(8, []) ] );
    GNode(4, [ GNode(9, [ GNode(12, [ GNode(13, []);
                                      GNode(14, []) ]) ]);
               GNode(10, []) ])
  ])
```

Notez que le constructeur `Empty` est ici inutile. En fait, ce constructeur est un peu parasite car on assimile les expressions `GNode(x,[])` et `GNode(x, [Empty])`. Il ne sert réellement que si l'arbre entier est vide. D'ailleurs, la littérature ignore en général l'arbre vide, dans le cas des arbres généraux.

La liste des sous-arbres attachée à un nœud est appelée *forêt*. Ainsi une arbre général est :

— soit vide ;
— soit composé d'une étiquette et d'une forêt ;

et une forêt est ;

— soit vide ;
— soit composée d'un arbre et d'une forêt.

Les arbres et les forêt sont ainsi *mutuellement définis*, de manière récursive. En conséquence, un schéma simple de traitement récursif des arbres généraux est le schéma de *récurrence mutuelle*. Par exemple la fonction de test d'appartenance d'un élément à un arbre est réalisée par une fonction sur les arbres associée à une fonction sur les forêts ; ces deux fonctions s'appelant l'une, l'autre :

```
let rec gtree_mem (z:'a) (gt : 'a gtree) : bool =
  match gt with
```

```
      GEmpty -> []
    | GNode(x, gts) -> (z=x) || (forest_mem z gts)
and forest_mem z (gts:('a gtree) list) : bool =
  match gts with
      [] -> []
    | gt::gts -> (gtree_mem z gt) || (forest_mem z gts)
```

La récurrence mutuelle est réalisée dans le langage avec les mots clé `let rec` (pour l'aspect «récursif») et `and` (pour l'aspect «mutuel»).

4 Structures modifiables et programmation impérative

Nous avons vu sur les listes la fonction `List.map` qui calcule, à partir d'une fonction `f` et d'une liste `xs` une *nouvelle liste* résultant de l'application de `f` à chaque élément de `xs`. Si l'on illustre l'empreinte mémoire du programme

```
let xs = [4;5;7;9] in
  (List.map succ xs)
```

on obtient quelque chose comme



avant l'évaluation de `(List.map succ xs)`, la liste `xs` est allouée en mémoire : *après* l'évaluation de `(List.map succ xs)` une nouvelle liste (anonyme) est allouée, sans que la première ait été modifiée.

Le résultat de `List.map` a été crée par *copie de la structure* de `xs`(allocation de 4 cellules). On voudrait pouvoir réaliser cela *en place*, c'est-à-dire, qu'après l'évaluation du schéma d'application, on ait en mémoire la situation suivante :



La structure initiale de `xs` a été conservée mais le contenu de ses cellules a été *modifié*. On a opéré l'application *en place*.

On n'obtiendra pas, en OCAML, exactement ce comportement, mais nous alons voir deux manières d'obtenir quelque chose d'approchant.

4.1 Le type `'a ref`

On trouve dans la bibliothèque standard (module `persvasive`) le type prédéfini `'a ref` avec les opérateurs suivants :

`ref` de type `'a -> 'a ref` qui crée une valeur de type `'a ref` : une *référence*, comme un *«pointeur vers»* ;

- **`(!)`** de type `'a ref -> 'a` opérateur de *déréférencement* ;
- **`(:=)`** de type `'a -> 'a ref -> unit`, opérateur (infixe) de *modification* de la valeur référencées, conne une affectation.

À titre d'illustration, considérer le code suivant et le résultat de son évaluation (sous la boucle d'interaction `ocaml` :

```
# let x = ref 42 in
  Printf.printf"Avant: %d\n" !x;
  x := !x+42;
  Printf.printf"Après: %d\n" !x
  ;;
Avant: 42
Après: 84
- : unit = ()
```

Un «map» en place On peut obtenir le schéma d'application en place en utilisant des lists de type `('a ref) list`. Ainsi les valeurs des cellules deviennent, en un sens, modifiables :

```
let rec mapset (f:'a->'a) (xs:('a ref) list) : unit =
  match xs with
    [] -> ()
  | x::xs -> (
     x := (f !x);
     mapset f xs)
```

Notez qu'ici la fonction `f` ne peut être de type `'a -> 'b` puisque les résultats de son application sont réaffecté à la liste `xs` de type `('a ref) list`.

On peut définir `mapset` en utilisant l'itérateur `List.iter` qui est de type `('a -> unit) -> 'a list -> unit` :

```
let mapset (f:'a -> 'a)  (xs:('a ref) list) : unit =
  List.iter (fun x -> x := (f !x)) xs
```

4.2 Enregistrements et champs modifiables

Le programmeur peut définir des types d'*enregistrement* (les `struct` de C). Par exemple

```
type point = {
  asbisse : int;
  ordonnee : int
}
```

La création d'une valeur de ce type s'écrit :

```
{ asbisse=$e_1$; ordonnee=$e_2$ }
```

où $e - 1$ et e_2 sont des expressions de type `int`. Si `x` est une valeur de type `point`, la notation pointée donne accès à ses champs : `x.absisse`, `x.ordonnee`.

Les champs, tout ou partie, d'un enregistrement peuvent être déclarés *modifiable*. On utilise pour cela le mot clef `mutable` :

```
type point = {
  mutable abscisse : int;
```

```
  mutable ordonnee : int
}
```

Dans ce cas, un *opérateur d'affectation*, noté `<-`, permet de modifier la valeur associée aux champs de la structure :

```
x.abscisse <- x.abscisse+42
```

En fait, le type `ref` est défini de cette manière :

```
type 'a ref = {
  mutable contents : 'a;
}
```

Listes chaînées modifiables en place

```
type 'a cell =
  mutable elt : 'a;
  mutable next : 'a clist

and 'a clist =
  Nil
| Cons of 'a cell
```

Exemple :

```
Cons { elt=1; next = Cons { elt=2; next= Cons { elt=3; next=Nil }}}
```

Le schèma d'application *en place* est définit de cette manière :

```
let rec mapset (f:'a -> 'a) (xs:'a clist) : unit =
  match xs with
    Nil -> ()
  | Cons c -> (
      c.elt <- (f c.elt);
      mapset f c.next)
```

Files d'attente avec modification *en place*.

On réalise des files d'attentes avec une structure qui donne une référence (un «pointeur») sur le premier et le dernier élément d'une `'a clist` :



On obtient cela en déclarant :

```
type 'a queue = {
  mutable first : 'a clist;
  mutable last : 'a clist
}
```

Création d'une file vide : les deux champs référencent une liste vide

```
let create () : 'a queue =
   first=Nil; last=Nil
```

Ajout d'un élément : on ajoute sur les derniers éléments de la lsite (champ `last`), avec un cas particulier lorsqu'il s'agit du permier élément ajouté

```
let add (x:'a) (xs:'a queue) : unit =
  let cc = Cons  elt=x; next=Nil  in
    match xs.last with
        Nil -> (
          xs.first <- cc; xs.last <- cc
        )
      | Cons c -> (
          c.next <- cc;
          xs.last <- cc
        )
```

Retrait d'un élément : on retire le premier élément de la liste (non vide) avec un cas particulier lorsqu'il s'agit de l'unique élément de la liste

```
let pop (xs:'a queue) : 'a =
  match xs.first with
      Nil -> failwith "empty queue"
    | Cons c -> (
let x = c.elt in
  xs.first <- c.next;
  if xs.first=Nil then xs.last <- Nil;
  x
      )
```

Notez que nos fonctions ne construisent jamais de files «pathologique» où l'un des champs serait `Nil` alors que l'autre serait un `Cons`.

Listes circulaires (non vides)

Les listes circulaires sont des structures bien adaptées à l'implantation de files d'attente :



Ici, c'est le champ `next` du dernier élément qui indique quel est le premier.

On se donne la structure suivante :

```
type 'a cell = {
  elt : 'a;
  mutable next : 'a cell
}
```

Pour créer une structure ciculaire, on utilise une définition récursive

```
let new_cell (x:’a) : ’a cell =
  let rec c =  elt=x; next=c  in c
```

À la création, le champ `next` de la structure pointe sur la structure elle-même.

On définit la fonction d’ajout de manière à réaliser l’ajout selon la discipline des filles d’attentes. Le résultat de la fonction est le nouveau «dernier» élément :

```
let add_cell (x:’a) (xs:’a cell) : ’a cell =
  let xs’ =  elt=x; next=xs.next  in
    xs.next <- xs’;
    xs’
```

La fonction de retrait donne la valeur du «premier» élément de la liste et elle supprime la cellule le contenant par effet de bord :

```
let rem_cell xs =
  let x = xs.next.elt in
    xs.next <- xs.next.next;
    x
```

Sur la base de cette structure, nous allons bâtir une nouvelle version des fonctionnalités des listes d’attentes en veillant à conserver le caractère de modification *en place* de `add` et `pop`. Il y a 2 sortes de modifications à considéer : le passage de la file vide à une file non vide (et vice-versa) ; l’ajout d’un nouvel élément.

Pour avoir la dualité de valeurs file vide et file non vide, on redéfinit le type `’a clist` :

```
type ’a clist =
    Nil
  | Cons of ’a cell ref
```

Le constructeur `Cons` a pour paramètre une *référence* pour permettre sa réaffectation après une opération d’ajout.

Le type des files d’attente est définit par :

```
type ’a queue = ’a clist ref
```

La référence permet le passage *en place* entre file vide et file non vide.

Création d’une file d’attente vide : référence vers une liste vide

```
let create () : ’a queue = (ref Nil)
```

Ajout d’un élément : on considère 2 cas
- si la file est vide, on crée la structure circulaire et on l’affecte à la file ;
- si une structure circulaire existe, on y ajoute une nouvelle cellule qui devient la valeur de la structure.

```
let add (x:’a) (xs:’a queue) : unit =
  match !xs with
      Nil -> xs := Cons (ref (new_cell x))
    | Cons c -> c := add_cell x !c
```

Notons ici que le «`Cons`» ne bouge pas, seule la référence qu’il encapsule est modifiée.

Retrait d’un élément : cette fonction doit tenir compte du cas particulier de la file contenant un seul élément.

Ce cas correspond à celui où le champs `next` de la structure circulaire «pointe» vers la structure elle-même (voir la fonction `new_cell`). Pour détecter ce cas, nous ne pouvons utiliser le test d’égalité usuel en écrivant `c = c.next`. En effet, la fonction `(=)` implémente un *test d’égalité structurelle*. Lorsqu’elle est appliquée à enregistrements, elle décompose la structure pour tester l’égalité des champs de la structure. Ici, calculer `c`

= `c.next` revient à calculer `c.elt = c.next.elt` et `c.next = c.next.next`. Mais si `c` ne contient qu'un seul élément alors la valeur de `c.next` est `c` elle-même et calculer `c.next = c.next.next` revient à calculer `c = c.next`. On a mis le pied dans un cercle vicieux dont on ne peut sortir.

Pour éviter ce piège, on utilise une autre sorte d'égalité : l'égalité dite *physique*. Lorsque les valeurs comparées sont des structures de données (type somme, n-uplet ou enregistrements), l'égalité physique se contente de vérifier l'identité d'allocation des structures. C'est exactement ce qu'il nous faut.

Lorsque l'élément à retirer est l'unique élément de la file, on remet celle-ci à `Nil` :

```
let pop (xs:'a queue) : 'a =
  match !xs with
      Nil -> failwith "empty queue"
    | Cons c -> (
        if (!c == !c.next) then (
          xs := Nil;
          !c.elt
        )
        else rem_cell c.contents
      )
```

4.3 Un module pour les files d'attente

Créer un *module* pour les files d'attente permet

1. Empaqueter la définition de la structure de listes d'attente dans une *unité de compilation*.
2. Limiter les fonctionalités fournies par le module au strict nécessaire : *type abstrait*.

On dispose en Ocaml de deux compilateurs :
- `ocamlc` : code-octet pour la machine virtuelle (*cf.* `ocamlrun`) ;
- `ocamlopt` : code natif (dépend de l'architecture cible).

On construit un module (simple) avec deux éléments
- une *interface*, source : extension `.mli`, version compiliée, extension `.cmi`.
- une *implémentation*, source : extension `.ml`, code octet, extension `.cmo`, code natif, extension `.cmx`

Nous allons construire un module pour les files d'attente basé sur leur implémentation avec les listes circulaires.

Fichier d'interface `queue.mli` Il contient les *déclarations* des constituant publiés du module :

```
type 'a t

exception Empty

val make : unit -> 'a t

val add : 'a -> 'a t -> unit

val pop : 'a t -> 'a
```

Les déclarations de fonctions sont introduites par le mot clef `val`.

Notez bien que le type des files d'attentes (`'a t`) est déclaré sans que les détails de sa réalisation soient données (types `'a cell`, `'a clist`, `'a clist ref`). C'est ce qui fait le caractère *abstrait* de la strucutre fournie par le module.

Notez également que les fonctions de manipulation des `'a cell` ne sont pas mentionnées. Elles seront ainsi inaccessibles par l'utilisatuer du module.

Fichier d'implémentation `queue.ml` Il contient, au minimum, les *définitions* des composants déclarés dans l'interface :

```
type ’a cell = {
  elt : ’a;
  mutable next : ’a cell
}

let new_cell (x:’a) : ’a cell =
  let rec c =  elt=x; next=c  in c

let add_cell (x:’a) (xs:’a cell) : ’a cell =
  let xs’ =  elt=x; next=xs.next  in
    xs.next <- xs’;
    xs’

let rem_cell xs =
  let x = xs.next.elt in
    xs.next <- xs.next.next;
    x

type ’a clist =
    Nil
  | Cons of ’a cell ref

type ’a t = ’a clist ref

exception Empty

let create () = (ref Nil)

let add x xs : unit =
  match !xs with
      Nil -> xs := Cons (ref (new_cell x))
    | Cons c -> c := add_cell x !c

let pop (xs:’a t) : ’a =
  match !xs with
      Nil -> raise Empty
    | Cons c -> (
        if (!c == !c.next) then (
          xs := Nil;
          !c.elt
        )
        else rem_cell c.contents
      )
```

Compilation et utilisation du module

```
ocamlc queue.mli
ocamlc -c queue.ml
```

On obtient les fichiers `queue.cmi` et `queue.cmo` qui définissent les module `Queue` (notez la majuscule).

Utilisation du module avec la boucle `ocaml` :

```
        OCaml version 4.02.3

# #load "queue.cmo" ;;
# Queue.create ;;
    unit -> Queue.t
```

Notez que le nom du type ainsi que le nom des fonctions sont précédés du nom du module. Il en va de même pour l'exception : `Queue.Empty`.

Supposons le fichier source d'un programme qui utilise les fonctionnalités du module `Queue`. Appelons le `prog.ml`. Pour compiler ce programme :

```
ocamlc -o prog queue.cmo prog.ml
```

On obtient un exécutable (code-octet) appelé `prog`.

4.4 Les tableaux et boucles

Les tableaux sont des *structures linéaires statiques*. Ils sont linéaires car les données y sont rangées selon un ordre défini par un indice de position. L'indice du premier élément d'un tableau est 0 ; celui du deuxième est 1, etc. Les valeurs contenues dans un tableau sont rangées consécutivement en mémoire. C'est cette structure qui permet l'accès aux éléments du tableau par leur indice. En effet, avec une telle disposition, l'indice suffit à déterminer la position (adresse), en mémoire, de l'élément correspondant. En général, et c'est le cas en OCAML, les tableaux sont des structures statiques, en ce sens que leur taille (nombre d'éléments qu'ils peuvent contenir) est donné à leur construction et ne peut être modifiée.

En revanche, la valeur des cellules d'une tableau est modifiable.

Le type des tableau est un type paramétré : `'a array`. Le module `Array` fournit les primitives de manipulation des tabelaux :

`Array.make` de type `int -> 'a -> 'a array` : `(Array.make len x)` contruit un tableau de longueur `len` dont les cellules sont initialisées avec la valeur de `x`. Il n'y a qu'un seul tableau de type `'a array` : le tableau vide, dont on ne peut pas faire grand chose.

`Array.get` de type `'a array -> int -> 'a` : `(Array.get xs i)` donne la valeur, il elle existe, de l'élément de `xs` dont l'indice est la valeur de `i`. Si une telle valeur n'exsite pas, l'exception `Invalid_argument "index out of bounds"` est déclenchée.

L'écriture `xs.(i)` est un racourci pour `(Array.get xs i)`.

`Array.set` de type `'a array -> int -> 'a -> unit` : `(Array.set xs i x)` donne la valeur de `v` à l'élement de `xs` dont l'indice est la valeur de `i`, si celui-ci existe. L'exception l'exception `Invalid_argument "index out of bounds"` est déclenchée sinon.

L'écriture `xs.(i) <- v` est un raccourci pour `(Array.set xs i x)`.

`Array.length` de type `'a array -> int` : `(Array.length xs)` donne la taille du tableau `xs`.

On peut réaliser un schéma d'application en place sur les tableaux de la manière suivante :

```
let mapset (f:'a -> 'a) (xs:'a array) : unit =
  let rec loop i =
    if (i < Array.length xs) then (
```

```
        xs.(i) <- (f xs.(i));
        loop (i+1)
      )
    in
      (loop 0)
```

Notez ici l'usage d'un `if` *unilatère* (sans `else`). Ceci est permis ici car l'expression (ici, une séquence) associée au `then` est de type `unit`. Si nous insistons sur le *ici*, c'est que c'est la seule configuration où l'usage d'un `if` unilatère est autorisée en OCAML.

On peut aussi utiliser un combinateur fourni par le module `Array` : `Array.iteri`, de type `(int -> 'a -> unit) -> 'a array -> unit`. On écrit alors :

```
let mapset (f:'a -> 'a) (xs:'a array) : unit =
  Array.iteri (fun i x -> xs.(i) <- (f x)) xs
```

L'argument `x` de la fonction itérée reçoit la valeur de l'élément du tableau dont l'indice est la valeur reçue par l'argument `i`. Ainsi, l'affectation `xs.(i) <- (f x)` de la fonction itérée produit le même effet que l'affectation `xs.(i) <- (f xs.(i))` de notre version précédente de `mapset`.

Mais, bien entendu, la manière la plus spontatnée que l'on a de définir une telle fonction est l'utilisation d'une *boucle* *`for`* :

```
let mapset (f:'a -> 'a) (xs:'a array) : unit =
  for i=0 to (Array.length xs)-1 do
     xs.(i) <- (f xs.(i))
  done
```

Itération non bornée Nous avons vu comment la fonction `find` de recherche dans une liste d'un élément réclamait l'"implantation d'une itération *non bornée*. Il y a plusiseurs manière de réaliser cela lorque l'on cherche dans un tableau.

La première est d'utiliser une boucle récursive sur les indices :

```
let find (p:'a -> bool) (xs:'a array) : 'a =
  let len = Array.length xs in
  let rec loop i =
    if (i < len) then
      if (p xs.(i)) then xs.(i)
      else (loop (i+1))
    else
      raise Not_found
  in
    (loop 0)
```

La deuxième est d'utiliser une boucle `while` avec un indice modifiable :

```
let find (p:'a -> bool) (xs:'a array) : 'a =
  let len = Array.length xs in
  let i = ref 0 in
    while (!i < len) && not (p xs.(!i)) do
      i := !i+1
    done;
    if (!i < len) then xs.(!i)
    else raise Not_found
```

La troisième est d'utiliser une itération bornée avec la boucle `for`, mais dont on s'échappe grace au mécanisme des exceptions :

```
exception Found_at of int
let find (p:'a -> bool) (xs:'a array) : 'a =
  let len = Array.length xs in
    try
      for i=0 to len-1 do
        if (p xs.(i)) then raise (Found_at i)
      done;
      raise Not_found
    with
      Found_at i -> xs.(i)
```

On utilise ici le mécanisme d'exception pour, potentiellemnt, interrompre l'itération de la boucle `for`. L'interruption est réalisée avec une exception définie (`Found_at`) qui véhicule une valeur – ici, un indice. Deux choses peuvent se produire au cours de la boucle `for` :

- ou bien celle-ci arrive à son terme : alors l'expression `raise Not_found` est atteinte, et c'est le résultat de la fonction ;
- ou bien un élément du tableau satisfait `p` et la boucle est interrompue par `raise (Found_at i)`. Dans ce cas, la continuation de l'évaluation est passée à l'analyse de cas d'exception (filtrage) qui suit le mot clef `with` et, puisque l'exception `Found_at` a été déclenchée, on donne la valeur de l'élément du tableau qui a statisfait `p` dont l'indice nous est transmis par l'exception.

5 Types et structures de données

Pour représenter les structures d'arbres généraux, nous avons défini le type suivant

```
type 'a gtree =
  GEmpty
| GNode of 'a * ('a gtree) list
```

Cette *représentation* souffre toutefois d'un vice : il existe une infinité de manières de réprésenter les arbres contenant un seul élément :

```
GNode(x,[])
GNode(x,[Empty])
GNode(x,[Empty; Empty])
..
GNode(x,[Empty; ..;Empty])
..
```

On peut palier ce défaut en séparant les arbres vides des arbres non vides. De cette manière, un arbre non vide ne pourra contenir d'arbre vide perturbateur.

```
type 'a gtree1 = GNode1 of 'a * 'a gtree1
type 'a gtree =
  Empty
| GNode of 'a gtree1
```

ou, de manière équivalente

```
type 'a gtree1 = GNode1 of 'a * 'a gtree1
type 'a gtree = ('a gtree) option
```

Mais on définit ou utilise alors deux types pour une seule structure.

Une autre manière de palier ce défaut est de définir la structure comme un *type abstrait* au moyen d'un module. On pose la signature (`agtree.mli`

```
type 'a t
exception Empty
val empty: unit -> 'a t
val gnode: 'a -> ('a t) list -> 'a t
val is_empty: 'a t -> bool
val get_label: 'a t -> 'a
val get_list: 'a t -> ('a t) list
```

Les fonctions `empty` et `gnode` jouerons le rôle des constructeurs des valeurs du type `'a AGtree.t`. C'est dans leur implémentation, en particulier, celle de `gnode` que nous prendrons garde à ne pas introduire de représentation non désirées.

```
type 'a t =
    TEmpty
  | TNode of ('a * ('a t) list)
exception Empty
let empty() = TEmpty
let gnode x gts = TNode (x, List.filter (fun gt -> gt<>TEmpty) gts)
let is_empty gt = (gt = TEmpty)
let get_label gt =
  match gt with
      TEmpty -> raise Empty
    | TNode(x,_) -> x
let get_list gt =
  match gt with
      TEmpty -> raise Empty
    | TNode(_,gts) -> gts
```

Mais on perd avec cela les facilités du filtrage. On y supplée en définissant un *reconnaisseur* d'arbre vide (`is_empty`) et deux *accesseurs* pour décomposer les arbres non vides (`get_label` et `get_list`).

Ainsi, une structure de donnée n'est pas nécessairement immédiatement définissable comme un *type de données*. Nous allons nous pencher mainteant sur un cas exemplaire de cette situation : les graphes.

5.1 Les graphes (orientés)

Un graphe est une structure théoriquement simple, mais dont l'implantation n'est pas immédiate.

Théoriquement, un graphe est simplement une *relation binaire* sur un ensemble de valeurs. Si A est notre ensemble de valeurs, un graphe d'éléments de A est donc une liste de couples de $A \times A$.

On représente souvent les graphes graphiquement :



Les éléments contenus dans le graphes sont appelés *sommet* (*vertex* en anglais). Les flèches qui les relient sont appelées *arcs* (*edge* en anglais).

5.2 Représentation simpliste des graphes

Si l'on se base sur la définition simple des graphes (listes de couples) on peut représenter le graphe dessiné ci-dessus par la liste

```
[ (1,2); (1,3); (2,4); (3,2); (3,5); (3,6); (5,2); (5,4); (6,5) ]
```

qui est de type `(int * int) list`.

Voisinage On dit qu'un sommet v_2 est *voisin* d'un sommet v_1 si il existe un arc de v_1 vers v_2. C'est-à-dire que (v_1, v_2) est un arc du graphe. On dit également que v_2 est *adjacent* à v_1.

On peut définir cette propriété de la manière suivante :

```
let exists_edge g v1 v2 =
  List.exists (fun x -> x=(v1,v2)) g
```

On peut également calculer la liste des voisins d'un sommet :

```
let adj_list g v =
  List.map snd ((List.filter (fun (v',_) -> v'=v)) g)
```

ou, de manière plus directe

```
let adj_list g v =
  List.fold_left (fun adjs (v1,v2) -> if (v1=v) then (v2::adjs) else adjs) [] g
```

Chemin Un *chemin* entre deux sommets d'un graphe est une suite d'arcs reliant l'un à l'autre. Dans notre exemple, il y a un chemin de 1 à 4 qui passe par 2 (il y en a un autre qui passe par 3 puis 5).

Pour savoir s'il existe un chemin d'un sommet v_1 vers un sommet v_2 on applique la méthode récursive suivante :

- ou bien il existe un arc entre v_1 et v_2
- ou bien, il existe un voisin de v de v_1 tel q'il existe un chemin de v à v_2. C'est-à-dire, il existe un sommet v parmi les voisins de v_1 tel qu'il existe un chemin de v vers v_2.

On peut *paraphraser* cette défintion en la définition OCAML suivante :

```
let rec exists_path g v1 v2 =
  (exists_edge g v1 v2)
  || (List.exists (fun v -> (exists_path g v v2)) (adj_list g v1))
```

Cette méthode simple fonctionne, à condition qu'il n'y ait jamais, dans le graphe, de chemin entre un sommet et lui-même (ce que l'on appelle un *cycle*). En effet, si l'on considère le graphe `g` représenté par la liste `[ (1,1); (1,2); (2,2); (2,3) ]` et que l'on veuille la valeur de `(exists_path g 1 3)` on n'obtiendra pas de réponse (en fait, on obtient `Stack overflow during evaluation (looping recursion?)`).

Remarque : ici, nous n'avons pas eu de chance. Un autre arrangement dans la liste qui représente la graphe nous aurait donné la réponse (par exemple `[(1, 2); (2, 2); (2, 3); (1, 1)]`).

Si l'on veut se prémunir contre le danger de s'enfermer dans un cycle, il nous faut un moyen de savoir si l'algorithme a déjà visité ou non un sommet. On reformule ainsi l'agorithme de recherche :

- v_1 n'a pas été visité
- et

— ou bien il existe un arc entre v_1 et v_2

— ou bien, il existe un sommet v parmi les voisins de v_1 tel qu'il existe un chemin de v vers v_2.

Pour savoir si un sommet a été visité ou non, on peut simplement maintenir à jour une liste. On obtient ainsi la définition

```
let exists_path g v1 v2 =
  let rec loop v visited =
    (not (List.mem v visited))
    && ((exists_edge g v v2)
        || (List.exists (fun v' -> (loop v' (v::visited)))
                        (adj_list g v)))
  in (loop v1 [])
```

ou, dans un style plus impératif :

```
let exists_path g v1 v2 =
  let visited = ref [] in
  let rec loop v =
    (not (List.mem v1 !visited))
    && ((List.mem v2 (adj_list g v1))
        || (visited := v1::!visited;
            List.exists (fun v' -> loop v') (adj_list g v)))
  in
     (loop v1
```

5.3 Listes d'adjacences

Si notre représentation simpliste des graphes permet de définir les fonctions nécessaires, elle pêche en revanche beaucoup par ses performances : il faut revisiter tout le graphe chaque fois que l'on veut la listes des voisins d'un sommet. Or, il est tout-à-fait possible de précalculer cette information et d'associer à chaque sommet la liste de ses voisins, ce que l'on appelle une *liste d'adjacence*. On représenterait alors le graphe comme une *liste d'association* entre les sommets et la lsite de leur voisin. Pour notre exemple, cela donne :

```
[ (1, [2; 3]); (2, [4]); (3, [2; 5; 6]); (5, [2; 4]); (6, [5]) ]
```

On peut gagner encore un peu, si, à la place des listes d'association, on utilise une structure de *table d'association*.

Tables de hachage Les *structures associatives* ont pour rôle d'associer des *valeurs* à des *clé*. Si `k` est le types des clefs et `v` le type des valeurs, on peut réaliser des structures associatives en créant des listes de type `(k * v) list`. L'accès aux valeurs demande que la liste des clefs soit parcourue.

Une structure de tableau fournit un accès direct aux valeurs si l'on connaît leur position (indice) dans le tableau. En mettant en relation des clefs et des indices, on obtient des structures associatives pour lesquelles l'accès à la valeur associée à une clef coûte uniquement le temps de calcul de l'indice correspondant à une clef. Ce temps pet être constant pour toute clef.

Les *fonctions de hachage* sont des fonctions de bas niveau qui ont pour rôle de ramener la représentation mémoire d'une valeur à la taille de la représentation d'un entier. Cet entier peut servir de base au calcul d'un indice dans un tableau. Ainsi, on peut, à l'aide d'un tableau, obtenir une structure associative.

La bibilothèque standard du langage OCAML fournit le module `Hashtbl` qui fournit à son tour les principales fonctions de manipulation de tables de hachage. En utilisant ce module, pour représenter la structure associative d'un graphe, on rédéfinit nos fonctions de base pour les graphes de la manière suivante :

```
type 'a graph = ('a, 'a list) Hashtbl.t

let create (n:int) : ('a graph) = Hashtbl.create n

let add_vertex (g:'a graph) (v:'a) : unit =
  Hashtbl.add g v []

let add_edge (g:'a graph) (v1:'a) (v2:'a) : unit =
  Hashtbl.replace g v1 (v2::(Hashtbl.find g v1))

let adj_list (g:'a graph) (v:'a) : ('a list) =
  Hashtbl.find g v

let exists_edge (g:'a graph) (v1:'a) (v2:'a) : bool =
  List.mem v2 (adj_list g v1)
```

L'usage du module `Hashtbl` induit un caractère impératif pour la définition des fonctions de construction du graphe `add_vertex` et `add_edge`. Notez aussi que la création du graphe demande un paramètre entier qui est utilisé comme indication de la taille de la table à créer. Il doit idéalement avoir pour valeur le nombre de sommets du graphe.